Key Homework 17

Math 277, Fall 2005

Ordinary Differential Equations

Dr. Goutziers

Initializations

```
> restart;
  with(PDEtools):
>
```

41: Page 179

To solve the differential equation

$$x^2\left(\frac{d^2}{dx^2}\,y(x)\right)+2\,x\left(\frac{d}{dx}\,y(x)\right)-6\,y(x)=0$$

we use the transformation $x=\mathbf{e}^t$ translate it into a linear differential equation with constant coefficients.

Code the differential equation and the transformation.

```
> deq_x:=x^2*diff(y(x), x$2)+2*x*diff(y(x), x)-6*y(x)=0;
  tr:=x=exp(t);
```

$$deq\_x := x^2\left(\frac{d^2}{dx^2}\,y(x)\right)+2\,x\left(\frac{d}{dx}\,y(x)\right)-6\,y(x)=0$$

$$tr := x=\mathbf{e}^t$$

Transform the differential equation into a differential equation with independent variable t. The mathematical details of this process were presented in class and can be found on Maple Lesson 14.

```
> deq_t:=simplify(dchange(tr, deq_x));
```

$$deq\_t := \left(\frac{d}{dt}\,y(t)\right)+\left(\frac{d^2}{dt^2}\,y(t)\right)-6\,y(t)=0$$

Find the auxiliary equation and compute the eigenvalues.

```
> aux:=simplify(eval(subs(y(t)=exp(r*t), deq_t))/exp(r*t));
```

$$aux := r+r^2-6=0$$

```
> evals:=solve(aux, r);
```

$$evals := 2, -3$$

Construct the general solution of the differential equation with independent variable t.

```
> ygt:=add(c[k]*exp(evals[k]*t), k=1..2);
```

$$ygt := c_1\,\mathbf{e}^{(2\,t)}+c_2\,\mathbf{e}^{(-3\,t)}$$

Create the inverse transformation.

```
> invtr:=isolate(tr, t);
```

$$invtr := t=\ln(x)$$

Use the inverse transformation to find the general solution of the differential equation with independent variable x.

```
> ygx:=expand(subs(invtr, ygt));
```

$$ygx := c_1\,x^2+\frac{c_2}{x^3}$$

42: Page 179

To solve the differential equation

$$x^2\left(\frac{d^2}{dx^2}\,y(x)\right)-3\,x\left(\frac{d}{dx}\,y(x)\right)+6\,y(x)=0$$

we use the transformation $x = \mathbf{e}^t$ translate it into a linear differential equation with constant coefficients.

Code the differential equation and the transformation.

```
> deq_x:=x^2*diff(y(x), x$2)-3*x*diff(y(x), x)+6*y(x)=0;
  tr:=x=exp(t);
```

$$deq\_x := x^2 \left(\frac{d^2}{dx^2} y(x)\right) - 3\,x \left(\frac{d}{dx} y(x)\right) + 6\,y(x) = 0$$

$$tr := x = \mathbf{e}^t$$

Transform the differential equation into a differential equation with independent variable t. The mathematical details of this process were presented in class and can be found on Maple Lesson 14.

```
> deq_t:=simplify(dchange(tr, deq_x));
```

$$deq\_t := -4 \left(\frac{d}{dt} y(t)\right) + \left(\frac{d^2}{dt^2} y(t)\right) + 6\,y(t) = 0$$

Find the auxiliary equation and compute the eigenvalues.

```
> aux:=simplify(eval(subs(y(t)=exp(r*t), deq_t))/exp(r*t));
```

$$aux := -4\,r + r^2 + 6 = 0$$

```
> evals:=solve(aux, r);
```

$$evals := 2 + \sqrt{2}\,I, 2 - \sqrt{2}\,I$$

Note that this time the eigenvalues are complex, so the corresponding linearly independent solutions are of the form

$$\mathbf{e}^{(\alpha\,t)} \cos(\beta\,t) \quad \text{and} \quad \mathbf{e}^{(\alpha\,t)} \sin(\beta\,t)$$

Hence, the general solution of the differential equation with independent variable t is given by

```
> ygt:=c[1]*exp(Re(evals[1])*t)*cos(Im(evals[1])*t)+c[2]*exp(Re(evals[1])*t)*sin(Im(eval
  s[1])*t);
```

$$ygt := c_1\,\mathbf{e}^{(2\,t)} \cos(\sqrt{2}\,t) + c_2\,\mathbf{e}^{(2\,t)} \sin(\sqrt{2}\,t)$$

Create the inverse transformation.

```
> invtr:=isolate(tr, t);
```

$$invtr := t = \ln(x)$$

Use the inverse transformation to find the general solution of the differential equation with independent variable x.

```
> ygx:=expand(subs(invtr, ygt));
```

$$ygx := c_1\,x^2 \cos(\sqrt{2}\,\ln(x)) + c_2\,x^2 \sin(\sqrt{2}\,\ln(x))$$

```
>
```